

Testfragen und Aufgaben zu Betriebssystemaufrufen

1. .DATA

```
Meldung1 DB 'Ende des Beispielprogramms',13,10,'$'
Meldung2 DB 'Programm beendet',13,10
```

.CODE

```
Programmstart:          ; Label haben einen Doppelpunkt am Ende
    mov ax,@data         ; Uebergabe der Adresse des Datensegments
                        ; zur Laufzeit
    mov ds,ax            ; DS zeigt nun auf das Datensegment

;Alle folgenden Abschnitte enthalten je einen Fehler, finden Sie diese!

; ***** Abschnitt1 *****
mov dx,OFFSET Meldung1  ; Offset der Adresse des Strings
int 21h                 ; Interrupt 21h : Aufruf von DOS

; ***** Abschnitt2 *****
mov dl,'A'
mov ah,02h              ; Bildschirmausgabe mit DOS
int 21

; ***** Abschnitt3 *****
mov ah,03h              ; Funktion 3: Lies Cursorposition und -groesse
                        ; Rueckgabe ch,cl: erste und letzte Scanlinie
                        ;          dh,dl: Reihe und Spalte
mov bx,0                ; Bildschirmseite 0
int 10h                 ; Int 10h (Video)
mov cx,0
mov dx,ax

; ***** Abschnitt4 *****
mov ah,9                ; DOS-Funktion, die einen durch $ begrenzten
                        ; String auf den Bildschirm ausgibt
mov dx,OFFSET Meldung2  ; Offset der Adresse des Strings
int 21h                 ; Interrupt 21h : Aufruf von DOS

; ***** Abschnitt5 *****
; Programmende, die Kontrolle muss explizit an DOS zurueckgegeben werden
mov ah,04Ch             ; ah=04C : DOS-Funktion "terminate the program"
mov al,0                ; DOS-Return-Code 0
```

Abschnitt 1: Nummer des Funktionsaufrufes wurde nicht in AH hinterlegt.

Abschnitt 2: int 21 dezimal statt 21h wird aufgerufen, Achtung tückischer Fehler!

Abschnitt 3: Aufruf korrekt aber sinnlos: Die Ergebnisse in DX und CX werden überschrieben.

Abschnitt 4: Aufruf korrekt, führt aber zur unbeabsichtigten Ausgabe von Zeichen, da die Zeichenkette nicht mit \$-Zeichen begrenzt ist.

Abschnitt 5: Vorbereitung korrekt aber Betriebssystemaufruf (hier Int 21h) fehlt.

2. Schreiben Sie ein kleines Programmstück, das die Anzahl der angeschlossenen Drucker als 2-Bit-Zahl in CX hinterlegt!

Anleitung: Der Int 11h gibt Auskunft über angeschlossene Peripherie. Er gibt in Register AX ein Statuswort zurück, auf den Bits 14 und 15 steht die Anzahl der angeschlossenen Drucker.

Lösung:

```
int 11h
shr ax,14
mov cx,ax
```

3. Schreiben Sie ein Programm, das die DOS-Hauptversionsnummer ermittelt und eine entsprechende Meldung auf dem Bildschirm ausgibt. Beispiel „V5“. Benutzen Sie dazu die Funktion 30h, „Get DOS Version Number“ des DOS-Aufrufs (Int 21h):

DOS-Funktion 30h : DOS-Versionsnummer ermitteln

Vor Aufruf	AH=30h
Nach Aufruf	AL = Hauptversionsnummer AH = Nebenversionsnummer

Beachten Sie: Um einer Ziffer anzuzeigen, kann nicht einfach das Bitmuster ausgegeben werden, das diese Zahl binär darstellt. Es muss stattdessen das Bitmuster ausgegeben werden, das diesem Zeichen in der ASCII-Tabelle entspricht. So wird die Ziffer 1 durch das ASCII-Zeichen 31h, die Ziffer 2 durch ASCII-Zeichen 32h usw. dargestellt. Überlegen Sie wie man von der Zahl, die das Betriebssystem zurückliefert, auf das richtige ASCII-Zeichen kommen kann.

Zur Text- und Zeichenausgabe können die schon bekannten DOS-Aufrufe benutzt werden.

Lösung:

```
mov dl,'V' ; V in dl bringen
mov ah,2 ; Funktion Zeichenausgabe
int 21h ; DOS aufrufen
mov ah,30h ; Funktionsnummer
int 21h ; DOS aufrufen
add al,30h ; Hauptversionsnummer wandeln in ASCII-Zeichen
mov dl,al ; in dl bringen
mov ah,2 ; Funktion Zeichenausgabe
int 21h ; DOS aufrufen
```